

# *Properties and Generalizations of the Fibonacci Word Fractal*

## *Exploring Fractal Curves*

**José L. Ramírez**  
**Gustavo N. Rubiano**

This article implements some combinatorial properties of the Fibonacci word and generalizations that can be generated from the iteration of a morphism between languages. Some graphic properties of the fractal curve are associated with these words; the curves can be generated from drawing rules similar to those used in L-systems. Simple changes to the programs generate other interesting curves.

### ■ 1. Introduction

The infinite Fibonacci word,

$$\mathbf{f} = 0\ 100\ 101\ 001\ 001\ 010\ 010\ 100\ 100\ 101\ \dots$$

is certainly one of the most studied words in the field of combinatorics on words [1–4]. It is the archetype of a Sturmian word [5]. The word  $\mathbf{f}$  can be associated with a fractal curve with combinatorial properties [6–7].

This article implements *Mathematica* programs to generate curves from  $\mathbf{f}$  and a set of drawing rules. These rules are similar to those used in L-systems.

The outline of this article is as follows. Section 2 recalls some definitions and ideas of combinatorics on words. Section 3 introduces the Fibonacci word, its fractal curve, and a family of words whose limit is the Fibonacci word fractal. Finally, Section 4 generalizes the Fibonacci word and its Fibonacci word fractal.

## ■ 2. Definitions and Notation

The terminology and notation are mainly those of [5] and [8]. Let  $\Sigma$  be a finite alphabet, whose elements are called symbols. A word over  $\Sigma$  is a finite sequence of symbols from  $\Sigma$ . The set of all words over  $\Sigma$ , that is, the free monoid generated by  $\Sigma$ , is denoted by  $\Sigma^*$ . The identity element  $\epsilon$  of  $\Sigma^*$  is called the empty word. For any word  $w \in \Sigma^*$ ,  $|w|$  denotes its length, that is, the number of symbols occurring in  $w$ . The length of  $\epsilon$  is taken to be zero. If  $a \in \Sigma$  and  $w \in \Sigma^*$ , then  $|w|_a$  denotes the number of occurrences of  $a$  in  $w$ .

For two words  $u = a_1 a_2 \dots a_k$  and  $v = b_1 b_2 \dots b_s$  in  $\Sigma^*$ , denote by  $uv$  the concatenation of the two words, that is,  $uv = a_1 a_2 \dots a_k b_1 b_2 \dots b_s$ . If  $v = \epsilon$ , then  $u\epsilon = \epsilon u = u$ ; moreover, by  $u^n$  denote the word  $uu \dots u$  ( $n$  times). A word  $v$  is a subword (or factor) of  $u$  if there exist  $x, y \in \Sigma^*$  such that  $u = xvy$ . If  $x = \epsilon$ , then  $u = vy$  and  $v$  is called a prefix of  $u$ ; if  $y = \epsilon$ , then  $u = xv$  and  $v$  is called a suffix of  $u$ .

The reversal of a word  $u = a_1 a_2 \dots a_k$  is the word  $u^R = a_k a_{k-1} \dots a_1$  and  $\epsilon^R = \epsilon$ . A word  $u$  is a palindrome if  $u^R = u$ .

An infinite word over  $\Sigma$  is a map  $\mathbf{u} : \mathbb{N} \rightarrow \Sigma$ , written as  $\mathbf{u} = a_1 a_2 a_3 \dots$ . The set of all infinite words over  $\Sigma$  is denoted by  $\Sigma^\omega$ .

### Example 1

The word  $\mathbf{p} = (p_n)_{n \geq 1} = 0110101000101\dots$ , where  $p_n = 1$  if  $n$  is a prime number and  $p_n = 0$  otherwise, is an example of an infinite word. The word  $\mathbf{p}$  is called the characteristic sequence of the prime numbers. Here are the first 50 terms of  $\mathbf{p}$ .

**Table[If[PrimeQ[n], 1, 0], {n, 50}]**

{0, 1, 1, 0, 1, 0, 1, 0, 0, 0, 1, 0, 1, 0, 0, 0,  
1, 0, 1, 0, 0, 0, 1, 0, 0, 0, 0, 0, 1, 0, 1, 0, 0,  
0, 0, 0, 1, 0, 0, 0, 1, 0, 1, 0, 0, 0, 1, 0, 0, 0}

### Definition 1

Let  $\Sigma$  and  $\Delta$  be alphabets. A morphism is a map  $h : \Sigma^* \rightarrow \Delta^*$  such that, for all  $x, y \in \Sigma^*$ ,  $h(xy) = h(x)h(y)$ .

There is a special class of words with many remarkable properties, the so-called Sturmian words. These words admit several equivalent definitions (see, e.g. [5], [8]).

### Definition 2

Let  $\mathbf{w} \in \Sigma^\omega$ . Let  $P(\mathbf{w}, n)$ , the complexity function of  $\mathbf{w}$ , be the map that counts, for all integer  $n \geq 0$ , the number of subwords of length  $n$  in  $\mathbf{w}$ . An infinite word  $\mathbf{w}$  is a Sturmian word if  $P(\mathbf{w}, n) = n + 1$  for all integer  $n \geq 0$ .

For example,  $P(01101010001010, 5) = 9$ .

```
StringPartition[string_, n_] := Table[StringTake
[string, {i, i + n - 1}],
  {i, 1, StringLength[string] - (n - 1)}]

StringPartition["01101010001010", 5]

{01101, 11010, 10101, 01010,
 10100, 01000, 10001, 00010, 00101, 01010}

Subwords[string_, n_] :=
  Intersection[StringPartition[string, n]]

Subwords["01101010001010", 5]

{00010, 00101, 01000, 01010,
 01101, 10001, 10100, 10101, 11010}

complexity[string_, i_] := Length[Subwords[string, i]]

Table[complexity["01101010001010", i], {i, 0, 20}]

{1, 2, 4, 7, 8, 9, 9, 8, 7, 6, 5, 4, 3, 2, 1, 0, 0, 0, 0, 0, 0}
```

Since for any Sturmian word,  $P(\mathbf{w}, 1) = 2$ , Sturmian words have to be over two symbols. The word  $\mathbf{p}$  in example 1 is not a Sturmian word because  $P(\mathbf{p}, 2) = 4 \neq 3$ .

Given two real numbers  $\alpha, \beta \in \mathbb{R}$  with  $\alpha$  irrational and  $0 < \alpha < 1, 0 \leq \beta < 1$ , define the infinite word  $\mathbf{w} = w_1 w_2 w_3 \dots$  as  $w_n = \lfloor (n+1)\alpha + \beta \rfloor - \lfloor n\alpha + \beta \rfloor$ . The numbers  $\alpha$  and  $\beta$  are the slope and the intercept, respectively. This word is called mechanical. The mechanical words are equivalent to Sturmian words [5]. As a special case,  $\beta = 0$  gives the characteristic words.

### Definition 3

Let  $\alpha$  be irrational,  $0 < \alpha < 1$ . For  $n \geq 1$ , define  $w_\alpha(n) = \lfloor (n+1)\alpha \rfloor - \lfloor n\alpha \rfloor$  and  $\mathbf{w}(\alpha) = w_\alpha(1) w_\alpha(2) w_\alpha(3) \dots$ ; then  $\mathbf{w}(\alpha)$  is called a characteristic word with slope  $\alpha$ .

On the other hand, note that every irrational  $\alpha \in (0, 1)$  has a unique continued fraction expansion

$$\alpha = [0, a_1, a_2, a_3, \dots] = \frac{1}{a_1 + \frac{1}{a_2 + \frac{1}{a_3 + \dots}}},$$

where each  $a_i$  is a positive integer. Let  $\alpha = [0, 1 + d_1, d_2, \dots]$  be an irrational number with  $d_1 \geq 0$  and  $d_n > 0$  for  $n > 1$ . To the directive sequence  $(d_1, d_2, \dots, d_n, \dots)$ , associate a sequence  $(s_n)_{\{n \geq -1\}}$  of words defined by  $s_{-1} = 1$ ,  $s_0 = 0$ ,  $s_n = s_{n-1}^{d_n} s_{n-2}$ ,  $n \geq 1$ .

Such a sequence of words is called a standard sequence. This sequence is related to characteristic words in the following way. Observe that, for any  $n \geq 0$ ,  $s_n$  is a prefix of  $s_{n+1}$ , which gives meaning to  $\lim_{n \rightarrow \infty} s_n$  as an infinite word. In fact, one can prove that each  $s_n$  is a prefix of  $\mathbf{w}(\alpha)$  for all  $n \geq 0$  and  $\mathbf{w}(\alpha) = \lim_{n \rightarrow \infty} s_n$  [5].

### ■ 3. Fibonacci Word and Its Fractal Curve

#### Definition 4

*Fibonacci words are words over  $\{0, 1\}$  defined inductively as follows:  $f_0 = 1$ ,  $f_1 = 0$ , and  $f_n = f_{n-1} f_{n-2}$ , for  $n \geq 2$ . The words  $f_n$  are referred to as the finite Fibonacci words. The limit*

$$\mathbf{f} = \lim_{n \rightarrow \infty} f_n = 0\ 100\ 101\ 001\ 001\ 010\ 010\ 100\ 100\ 101\ \dots \quad (1)$$

*is called the Fibonacci word.*

It is clear that  $|f_n| = F_n$ , where  $F_n$  is the  $n^{\text{th}}$  Fibonacci number, recalling that the Fibonacci number  $F_n$  is defined by the recurrence relation  $F_n = F_{n-1} + F_{n-2}$  for all integer  $n \geq 2$  and with initial values  $F_0 = F_1 = 1$ . The infinite Fibonacci word  $\mathbf{f}$  is a Sturmian word [5]; exactly,  $\mathbf{f} = \mathbf{w}\left(\frac{1}{\phi^2}\right)$ , where  $\phi = \frac{1+\sqrt{5}}{2}$  is the golden ratio.

Here are the first 50 terms of  $\mathbf{f}$ .

$$\begin{aligned} & \text{Table} \left[ \text{IntegerPart} \left[ (n+1) \frac{1}{\text{GoldenRatio}^2} \right] - \right. \\ & \quad \left. \text{IntegerPart} \left[ n \frac{1}{\text{GoldenRatio}^2} \right], \{n, 1, 50\} \right] \\ & \{0, 1, 0, 0, 1, 0, 1, 0, 0, 1, 0, 0, 1, 0, 1, 0, \\ & \quad 0, 1, 0, 1, 0, 0, 1, 0, 0, 1, 0, 1, 0, 0, 1, 0, 0, \\ & \quad 1, 0, 1, 0, 0, 1, 0, 1, 0, 0, 1, 0, 0, 1, 0, 1, 0\} \end{aligned}$$

**Definition 5**

The Fibonacci morphism  $\sigma : \{0, 1\}^* \longrightarrow \{0, 1\}^*$  is defined by  $\sigma(0) = 01$  and  $\sigma(1) = 0$ .

The Fibonacci word  $\mathbf{f}$  satisfies  $\mathbf{f} = \lim_{n \rightarrow \infty} \sigma^n(1)$  and  $\sigma^n(1) = f_n$  for all  $n \geq 1$ .

```
FibonacciWord[n_] :=  
  Nest[StringReplace[#, {"0" -> "01", "1" -> "0"}] &, "1", n]
```

Here are the first nine finite Fibonacci words.

```
TableForm[Table[FibonacciWord[i], {i, 1, 9}]]
```

0
01
010
01001
01001010
0100101001001
010010100100101001010
01001010010010100101001001001
010010100100101001010010010100101001010

**Definition 6**

Let  $\Phi : \{0, 1\}^* \longrightarrow \{0, 1\}^*$  be the map such that  $\Phi$  deletes the last two symbols.

The following proposition summarizes some basic properties about the Fibonacci word.

**Proposition 1**

The Fibonacci word and the finite Fibonacci words satisfy:

1. The words 11 and 000 are not subwords of the Fibonacci word.
2. Let  $ab$  be the last two symbols of  $f_n$ . For  $n \geq 2$ ,  $ab = 01$  if  $n$  is even and  $ab = 10$  if  $n$  is odd.
3. The concatenation of two successive Fibonacci words is almost commutative; that is,  $f_n f_{n-1}$  and  $f_{n-1} f_n$  have a common prefix of length  $F_n - 2$ , for all  $n \geq 2$ .
4.  $\Phi(f_n)$  is a palindrome for all  $n \geq 2$ .
5. For all  $n \geq 6$ ,  $f_n = f_{n-3} f_{n-3} f_{n-6} l_{n-3} l_{n-3}$ , where  $l_n = \Phi(f_n)ba$ ; that is,  $l_n$  exchanges the two last symbols of  $f_n$ .

## □ The Fibonacci Word Fractal

The Fibonacci word can be associated with a curve using a drawing rule. A particular action follows on the symbol read (this is the same idea as that used in L-systems [9]). In this case, the drawing rule is called “the odd-even drawing rule” [7].

symbol	position of symbol	Draw a line forward, then
1	any	stay straight
0	even	turn left
0	odd	turn right

▲ **Table 1.** The odd-even drawing rule.

### Definition 7

The  $n^{\text{th}}$  Fibonacci curve, denoted by  $\mathcal{F}_n$ , is the result of applying the odd-even drawing rule to the word  $f_n$ . The Fibonacci word fractal is defined as

$$\mathcal{F} = \lim_{n \rightarrow \infty} \mathcal{F}_n.$$


The program LShow is adapted from [10] to generate L-systems.

```

LShow[lstring_String, Ldelta_: 90. Degree, size_: 400] :=
Module[
  {Lpos = {0., 0.}, Ltheta = 0.},
Graphics[
  Line[DeleteCases[Map[Switch[#, "+", Ltheta += Ldelta;,
"-", Ltheta -= Ldelta;, "F",
  Lpos += {Cos[Ltheta], Sin[Ltheta]},
"B", Lpos -= {Cos[Ltheta], Sin[Ltheta]}, _, Lpos += 0.] &,
Characters[lstring]], Null]], AspectRatio → Automatic,
  ImageSize → {size, size}]
]

```

Figure 1 shows an L-system interpretation of the odd-even drawing rule.



▲ **Figure 1.** Interpretation of the odd-even drawing rule.

```

FibonacciLOGOword[n_] := StringReplace[
  FibonacciWord[n],
  {"10" → "FF+", "01" → "F-F", "00" → "F-F+"}
]


```

Here are the curves  $\mathcal{F}_n$  for  $n = 9, \dots, 21$ .

```

Manipulate[
  LShow[FibonacciLOGOword[n], 90. Degree, 400],
  {{n, 11}, 7, 21, 1, Appearance → "Labeled"},
  SaveDefinitions → True
]

```



The next proposition about properties of the curves  $\mathcal{F}_n$  and  $\mathcal{F}$  comes directly from the properties of the Fibonacci word from Proposition 1. More properties can be found in [7].

**Proposition 2**

*The Fibonacci word fractal  $\mathcal{F}$  and the curve  $\mathcal{F}_n$  have the following properties:*

1.  $\mathcal{F}$  is composed only of segments of length 1 or 2.
2. The number of turns in the  $\mathcal{F}_n$  curve is the Fibonacci number  $F_{n-1}$ .
3. The  $\mathcal{F}_n$  curve is similar to the curve  $\mathcal{F}_{n-3}$ .
4. The curve  $\mathcal{F}_n$  is symmetric.
5. The  $\mathcal{F}_n$  curve is composed of five curves:  $\mathcal{F}_n = \mathcal{F}_{n-3} \mathcal{F}_{n-3} \mathcal{F}_{n-6} \mathcal{F}'_{n-3} \mathcal{F}'_{n-3}$ , where  $\mathcal{F}'_n$  is the result of applying the odd-even drawing rule to the word  $l_n$ .

The next figure shows the curve  $\mathcal{F}_{17}$  and the five curves; here  $\mathcal{F}_{17} = \mathcal{F}_{14} \mathcal{F}_{14} \mathcal{F}_{11} \mathcal{F}'_{14} \mathcal{F}'_{14}$ .

```


LMove2[z_String, δ_, pos_List] := Block[
  {x, y, θ, moves},
  {x, y} = pos[[1]];
  θ = pos[[2]];
  moves = {{x, y}};
  Map[
    Switch[#, "+", θ += δ; , "-", θ -= δ; , "F",
      {x, y} += {Cos[θ], Sin[θ]};
      AppendTo[moves, {x, y}];] &,
    Characters[z]
  ];
  moves
]

ColorFibonacci[n_, size_: 400] := Module[
  {c},
  c = LMove2[FibonacciLOGOword[n], 90. Degree,
    {{0, 0}, N[90 Degree]}] // Chop;
  f3 = Fibonacci[n - 3];
  f6 = Fibonacci[n - 6];
  Graphics[{
    Blue, Line[Take[c, {1, f3}]],
    Red, Line[Take[c, {f3, 2 f3}]],
    Black, Line[Take[c, {2 f3, 2 f3 + f6}]],
    Red, Line[Take[c, {2 f3 + f6, 3 f3 + f6}]],
    Blue, Line[Take[c, {3 f3 + f6, 4 f3 + f6}]]],
  ImageSize -> {size, size}]
]

```



```
Manipulate[ColorFibonacci[n],  
  {{n, 11}, 7, 21, 1, Appearance → "Labeled"},  
  SaveDefinitions → True]
```



## □ Some Variations

The Fibonacci word and other words can be derived from the dense Fibonacci word, which was introduced in [7].

### Definition 8

*The dense Fibonacci word  $\hat{\mathbf{f}}$  comes from the Fibonacci word  $\mathbf{f}$  by applying the morphism*

$$\eta(00) = 0, \eta(01) = 1, \eta(10) = 2, \quad (2)$$

*so that  $\hat{\mathbf{f}} = 10221022110211021102210221021 \dots$ .*

```
DenseFibonacciWord[n_] :=  
  StringReplace[FibonacciWord[n],  
    {"00" → "0", "01" → "1", "10" → "2"}]
```

```
DenseFibonacciWord[10]
```

```
102210221102110211022102211021102110221022101
```

Given a drawing rule, the global angle is the sum of the successive angles generated by the word through the rule. With the natural drawing rule,  $\Delta(1) = -\pi/2$ ,  $\Delta(0) = 0$ ,  $\Delta(2) = \pi/2$ , then  $\Delta(120) = \Delta(1) + \Delta(2) + \Delta(0) = 0$ .

For a drawing rule, the resulting angle of a word  $d$  is the function  $\Delta$  that gives the global angle. A morphism  $\theta$  preserves the resulting angle if for any word  $w$ ,  $\Delta(\theta(w)) = \Delta(w)$ ; moreover, a morphism  $\theta$  inverts the resulting angle if for any word  $w$ ,  $\Delta(\theta(w)) = -\Delta(w)$ .

The dense Fibonacci word is strongly linked to the Fibonacci word fractal because  $\hat{\mathbf{f}}$  can generate a whole family of curves whose limit is the Fibonacci word fractal [7]. All that is needed is to apply a morphism to  $\hat{\mathbf{f}}$  that preserves or inverts the resulting angle.

Here are some examples.

```

NewFibonacci[n_, string0_, string1_, string2_, angle_] :=
LShow[
  "+" <> StringReplace[
    StringReplace[DenseFibonacciWord[n],
      {"0" → string0, "1" → string1, "2" → string2}],
    {"0" → "F", "1" → "F-", "2" → "F+"}], N[angle Degree],
  150]


Grid[{
  Text /@ {
    "0 → ε, 1 → 1, 2 → 2",
    "0 → 12, 1 → 1, 2 → 2",
    "0 → 0, 1 → 1, 2 → 2"
  },
  {
    NewFibonacci[16, "", "1", "2", 90],
    NewFibonacci[16, "12", "1", "2", 90],
    NewFibonacci[16, "0", "1", "2", 90]
  },
  Text /@ {
    "0 → 21, 1 → 02, 2 → 10",
    "0 → 210, 1 → 020, 2 → 10",
    "0 → 102, 1 → 2, 2 → 1"
  },
  {
    NewFibonacci[16, "21", "02", "10", 90],
    NewFibonacci[16, "210", "020", "10", 90],
    NewFibonacci[16, "102", "2", "1", 90]
  }
}, Frame → All]

```

0 → ε, 1 → 1, 2 → 2	0 → 12, 1 → 1, 2 → 2	0 → 0, 1 → 1, 2 → 2
0 → 21, 1 → 02, 2 → 10	0 → 210, 1 → 020, 2 → 10	0 → 102, 1 → 2, 2 → 1

Here are some examples with other angles.

```
Grid[{
  Text /@ {
    "0 → 210, 1 → 020, 2 → 10",
    "0 → 2102, 1 → 020, 2 → 10",
    "0 → 210, 1 → 020, 2 → 10"
  },
  {
    NewFibonacci[16, "210", "020", "10", 100],
    NewFibonacci[16, "2102", "020", "10", 60],
    NewFibonacci[17, "210", "020", "10", 120]
  },
  Text /@ {
    "0 → 01, 1 → 2, 2 → 10",
    "0 → 01, 1 → 2, 2 → 10",
    "0 → 12, 1 → 1, 2 → 2"
  },
  {
    NewFibonacci[16, "01", "2", "10", 150],
    NewFibonacci[21, "01", "2", "10", 60],
    NewFibonacci[16, "12", "1", "2", 70]
  }
}, Frame → All]
```



## ■ 4. Generalized Fibonacci Words and Fibonacci Word Fractals

This section introduces a generalization of the Fibonacci word and the Fibonacci word fractal [11].

### Definition 9

The  $(n, i)$ -Fibonacci words are words over  $\{0, 1\}$  defined inductively by  $f_0^{[i]} = 0$ ,  $f_1^{[i]} = 0^{i-1} 1$ , and  $f_n^{[i]} = f_{n-1}^{[i]} f_{n-2}^{[i]}$ , for  $n \geq 2$  and  $i \geq 1$ . The infinite word

$$\mathbf{f}^{[i]} = \lim_{n \rightarrow \infty} f_n^{[i]}$$

is called the  $i$ -Fibonacci word.

The 2-Fibonacci word is the classical Fibonacci word. Here are the first six  $i$ -Fibonacci words.

```
PowerWord[n_, w_String] := Nest[# <> w &, w, n - 1]
```

```
iFibonacciWord[i_, 0] = "0";
iFibonacciWord[i_, 1] :=
  If[i > 1, PowerWord[i - 1, "0"] <> "1", "1"]
iFibonacciWord[i_, n_] :=
  iFibonacciWord[i, n - 1] <> iFibonacciWord[i, n - 2]
```

```
Text@
```

```
Column@
```

```
Table[Row[{Style["f", Bold] Row[{"", i, ""}], " = ",
  iFibonacciWord[i, 6], "..."}], {i, 1, 6}]
```

```
f[1] = 1011010110110...
```

```
f[2] = 010010100100101001010...
```

```
f[3] = 00100010010001000100100010010...
```

```
f[4] = 0001000010001000010000100010000100010...
```

```
f[5] = 000010000010000100000100000100001000001000010...
```

```
f[6] = 000001000000100000100000010000001000000100000010000010...
```

The following proposition relates the Fibonacci word  $\mathbf{f}$  to  $\mathbf{f}^{[i]}$ .

**Proposition 3**

Let  $\varphi_i : \{0, 1\}^* \rightarrow \{0, 1\}^*$  be the morphism defined by  $\varphi_i(0) = 0$  and  $\varphi_i(1) = 0^i 1$ ,  $i \geq 0$ ; then

$$\mathbf{f}^{[i+2]} = \varphi_i(\mathbf{f}) \quad (3)$$

for all  $i \geq 0$ .

**Definition 10**

The  $(n, i)$ -Fibonacci number  $F_n^{[i]}$  is defined recursively by  $F_0^{[i]} = 1$ ,  $F_1^{[i]} = i$ , and  $F_n^{[i]} = F_{n-1}^{[i]} + F_{n-2}^{[i]}$ , for all  $n \geq 2$  and  $i \geq 1$ .

The  $(n, 1)$ -Fibonacci numbers are the Fibonacci numbers and the  $(n, 2)$ -Fibonacci numbers are the Fibonacci numbers shifted by one. The following table shows the first terms in the sequences  $F_n^{[i]}$  and their reference numbers in the On-Line Encyclopedia of Sequences (OIES) [12].

```
text[i_] := Style["F", Italic]Style["n", Italic]Row[{"[", i, "]"}];
Text@
TableForm[
  Table[Table[LinearRecurrence[{1, 1}, {1, i}, n][[n]],
    {n, 1, 10}], {i, 1, 6}], TableHeadings -> {{
    Row[{text[1], "\tA000045"}],
    Row[{text[2], "\tA000045"}],
    Row[{text[3], "\tA000204"}],
    Row[{text[4], "\tA000085"}],
    Row[{text[5], "\tA022095"}],
    Row[{text[6], "\tA022096"}]
  }, Automatic]}
```

		1	2	3	4	5	6	7	8	9
$F_n^{[1]}$	A000045	1	1	2	3	5	8	13	21	34
$F_n^{[2]}$	A000045	1	2	3	5	8	13	21	34	55
$F_n^{[3]}$	A000204	1	3	4	7	11	18	29	47	76
$F_n^{[4]}$	A000085	1	4	5	9	14	23	37	60	97
$F_n^{[5]}$	A022095	1	5	6	11	17	28	45	73	118
$F_n^{[6]}$	A022096	1	6	7	13	20	33	53	86	139

**Proposition 4**

*The  $i$ -Fibonacci word and the  $(n, i)$ -Fibonacci word satisfy the following:*

1. The word 11 is not a subword of the  $i$ -Fibonacci word,  $i \geq 2$ .
2. Let  $ab$  be the last two symbols of  $f_n^{[i]}$ . For  $n \geq 1$ ,  $ab = 10$  if  $n$  is even and  $ab = 01$  if  $n$  is odd,  $i \geq 2$ .
3. The concatenation of two successive  $i$ -Fibonacci words is almost commutative; that is,  $f_{n-1}^{[i]} f_{n-2}^{[i]}$  and  $f_{n-2}^{[i]} f_{n-1}^{[i]}$  have a common prefix of length  $F_n^{[i]} - 2$  for all  $n \geq 2$  and  $i \geq 2$ .
4.  $\Phi(f_n^{[i]})$  is a palindrome for all  $n \geq 1$ .
5. For all  $n \geq 6$ ,  $f_n^{[i]} = f_{n-3}^{[i]} f_{n-3}^{[i]} f_{n-6}^{[i]} l_{n-3}^{[i]} l_{n-3}^{[i]}$ , where  $l_n^{[i]} = \Phi(f_n^{[i]}) b a$ .

**Theorem 1**

*Let  $\alpha = [0, i, \bar{1}]$  be an irrational number, with  $i$  a positive integer; then  $\mathbf{w}(\alpha) = \mathbf{f}^{[i]}$ .*

For the proof, see [11]. This theorem implies that  $i$ -Fibonacci words are Sturmian words.

Note that

$$[0, i, \bar{1}] = \frac{1}{i + \frac{1}{1 + \frac{1}{1 + \dots}}} = \frac{i - \phi}{i^2 - i - 1},$$

where  $\phi$  is the golden ratio.

□ **The  $i$ -Fibonacci Word Fractal**

**Definition 11**

*The  $(n, i)^{\text{th}}$  Fibonacci curve, denoted by  $\mathcal{F}_n^{[i]}$ , is the result of applying the odd-even drawing rule to the word  $f_n^{[i]}$ . The  $i$ -Fibonacci word fractal  $\mathcal{F}^{[i]}$  is defined as*


$$\mathcal{F}^{[i]} = \lim_{n \rightarrow \infty} \mathcal{F}_n^{[i]}.$$

Here are the curves  $\mathcal{F}_n^{[i]}$  for  $i = 2, 3, 4, 5, 6, 7$ .

```
d[i_, n_] :=
StringJoin[
Table[IntegerString[
Floor[(j + 1) * FromContinuedFraction[{0, i, {1}}]] -
Floor[(j) * FromContinuedFraction[{0, i, {1}}]]],
{j, 1, n}]]

iFibonacciFractal[i_, n_] :=
LShow[
"+" <> StringReplace[d[i, n],
{"00" → "F-F+", "01" → "F-F", "10" → "FF+"}],
90. Degree, 150]

Grid[
Partition[
Table[iFibonacciFractal[i,
LinearRecurrence[{1, 1}, {1, i}, 12][[12]]], {i, 2, 7}],
3], Frame → All]
```





**Proposition 5**

The  $i$ -Fibonacci word fractal and the curve  $\mathcal{F}_n^{[i]}$  have the following properties:

1. The Fibonacci fractal  $\mathcal{F}^{[i]}$  is composed only of segments of length 1 or 2.
2. The  $\mathcal{F}_n^{[i]}$  curve is similar to the curve  $\mathcal{F}_{n-3}^{[i]}$ .
3. The  $\mathcal{F}_n^{[i]}$  curve is composed of five curves:  

$$\mathcal{F}_n^{[i]} = \mathcal{F}_{n-3}^{[i]} \mathcal{F}_{n-3}^{[i]} \mathcal{F}_{n-6}^{[i]} \mathcal{F}_{n-3}'^{[i]} \mathcal{F}_{n-3}'^{[i]}.$$
4. The  $\mathcal{F}_n^{[i]}$  curve is symmetric.
5. The scale factor between  $\mathcal{F}_n^{[i]}$  and  $\mathcal{F}_{n-3}^{[i]}$  is  $1 + \sqrt{2}$ .

■ **Other Characteristic Words**

This section applies the above ideas to generate new curves from characteristic words (see Definition 3).


**Conjecture 1**

If  $\alpha = [0, a_1, \dots, a_n, \overline{1}]$ , then the curve displays the Fibonacci word fractal pattern.


```
CharacteristicFibonacciFractal[b_, n_] :=
LShow[
  "+" <> StringReplace[
    StringJoin[
      Table[IntegerString[Floor[(j + 1) b] - Floor[j b]],
        {j, n}]], {"00" → "F-F+", "01" → "F-F",
      "10" → "FF+"}], 90, Degree, 400]
```

Here are seven examples.


```
CharacteristicFibonacciFractal[  
  FromContinuedFraction[{0, 2, 2, 1, 2, 1, 2, {1}}],  
  33 000]
```




```
CharacteristicFibonacciFractal[
  FromContinuedFraction[{0, 9, 1, 3, {1}}], 44 000]
```




```
CharacteristicFibonacciFractal[  
  FromContinuedFraction[{0, 7, 7, 7, 7, {1}}], 35500]
```




```
CharacteristicFibonacciFractal[
  FromContinuedFraction[{0, 5, 10, 5, {1}}], 9900]
```




```
CharacteristicFibonacciFractal[  
  FromContinuedFraction[{0, 3, {5}}], 10 000]
```



```
CharacteristicFibonacciFractal[
  FromContinuedFraction[{0, 5, {2}}], 5000]
```



```
CharacteristicFibonacciFractal[  
  FromContinuedFraction[{0, 9, 3, 2, 1, {2, 3}}], 172 000]
```



## ■ Acknowledgments

The first author was partially supported by Universidad Sergio Arboleda under grant number USA-II-2012-14. The authors would like to thank Borut Jurčič-Zlobec from Ljubljana University for his help during the development of this article.



## ■ References

- [1] J. Cassaigne, “On Extremal Properties of the Fibonacci Word,” *RAIRO—Theoretical Informatics and Applications*, **42**(4), 2008 pp. 701–715. doi:10.1051/ita:2008003.
- [2] W. Chuan, “Fibonacci Words,” *Fibonacci Quarterly*, **30**(1), 1992 pp. 68–76. [www.fq.math.ca/Scanned/30-1/chuan.pdf](http://www.fq.math.ca/Scanned/30-1/chuan.pdf).
- [3] W. Chuan, “Generating Fibonacci Words,” *Fibonacci Quarterly*, **33**(2), 1995 pp. 104–112. [www.fq.math.ca/Scanned/33-2/chuan1.pdf](http://www.fq.math.ca/Scanned/33-2/chuan1.pdf).
- [4] F. Mignosi and G. Pirillo, “Repetitions in the Fibonacci Infinite Word,” *RAIRO—Theoretical Informatics and Applications*, **26**(3), 1992 pp. 199–204.
- [5] M. Lothaire, *Applied Combinatorics on Words (Encyclopedia of Mathematics and its Applications)*, Cambridge: Cambridge University Press, 2005.
- [6] A. Blondin Massé, S. Brlek, A. Garon, and S. Labbé, “Two Infinite Families of Polyominoes That Tile the Plane by Translation in Two Distinct Ways,” *Theoretical Computer Science*, **412**(36), 2011 pp. 4778–4786. doi:10.1016/j.tcs.2010.12.034.
- [7] A. Monnerot-Dumaine, “The Fibonacci Word Fractal,” preprint, 2009. [hal.archives-ouvertes.fr/hal-00367972/fr](http://hal.archives-ouvertes.fr/hal-00367972/fr).
- [8] J.-P. Allouche and J. Shallit, *Automatic Sequences: Theory, Applications, Generalizations*, Cambridge: Cambridge University Press, 2003.
- [9] P. Prusinkiewicz and A. Lindenmayer, *The Algorithmic Beauty of Plants*, New York: Springer-Verlag, 1990.
- [10] E. Weisstein. “Lindenmayer System” from Wolfram *MathWorld*—A Wolfram Web Resource. [mathworld.wolfram.com/LindenmayerSystem.html](http://mathworld.wolfram.com/LindenmayerSystem.html).
- [11] J. Ramírez, G. Rubiano, and R. de Castro, “A Generalization of the Fibonacci Word Fractal and the Fibonacci Snowflake,” 2013. arxiv:1212.1368v2.
- [12] OEIS Foundation, Inc. “The On-Line Encyclopedia of Integer Sequences.” (Aug 9, 2013) [oeis.org](http://oeis.org).

J. L. Ramírez and G. N. Rubiano, “Properties and Generalizations of the Fibonacci Word Fractal,” *The Mathematica Journal*, 2014. dx.doi.org/doi:10.3888/tmj.16-2.

## About the Authors

### José L. Ramírez

*Instituto de Matemáticas y sus Aplicaciones*  
*Universidad Sergio Arboleda*  
*Calle 74 no. 14 - 14 Bogotá, Colombia*  
[josel.ramirez@ima.usergioarboleda.edu.co](mailto:josel.ramirez@ima.usergioarboleda.edu.co)

### Gustavo N. Rubiano

*Departamento de Matemáticas*  
*Universidad Nacional de Colombia*  
*AA 14490, Bogotá, Colombia*  
[gnrubiano@unal.edu.co](mailto:gnrubiano@unal.edu.co)